Intermediate Social Statistics
Workshop 3
Kat Chzhen 
Topics: multinomial logit:  predicted probabilities and marginal effects; introduction to survival analysis.
Data sets:  ‘EQLS 2007_multinomial.dta’ from the European Quality of Life Survey 2007; ‘BHPS unemployment spells.dta’ from the British Household Panel Survey Combined Work-Life History Data, 1990-2005. 
Readings: Long, J.S. and Freese, J. (2006) Regression models for categorical dependent variables using Stata. Stata Press. Chapter 6.
Cleves, M., Gould, W. and Gutierrez, R. (2002) An introduction to survival analysis using Stata. Stata Press. 

INTRODUCTION
This week, we will focus on multinomial logistic regression models, calculating the effects of changes in the values of predictors on the estimated probabilities. We will then proceed to estimating simple parametric survival models.  

DATASETS
Copy the following datasets from the shared drive (Q:\Public\Labs_shared\Intermediate Social Statistics 2011) and save them in your own account: “EQLS 2007_multinomial.dta” and “BHPS unemployment spells.dta”.

ESTIMATING MULTINOMIAL LOGIT MODELS
Let’s use the European Quality of Life Survey to predict the frequency of internet usage in Great Britain, as a function of gender, age group, labour market activity, education, marital status, and presence of dependent children in the household. Open EQLS 2007_multinomial.dta and produce a frequency distribution of weekly internet use (internet) for Great Britain (country=15):

tab internet if country==15 [aw=wgt_target]

Have a look at the distribution of the predictor variables and estimate a multinomial logit model with category 1 (using internet every day) as the baseline category. Use probability weights with mlogit. The weight variable for single country analysis is wgt_target.

tab1  male agegroup activity kid edu urban married if country==15 
mlogit internet i.agegroup i.activity kid edu urban married if country==15 [pw=wgt_target]

The logit coefficients do not have a straightforward interpretation, but it is possible to comment on the direction and the statistical significance of the effects. The results suggest that, on average:
· People with higher education are less likely to use the internet “a couple of times a week”, “occasionally”, or “not at all”, as opposed to “every day”, than those with lower education, everything else being equal. 
· Unemployed and inactive people are less likely to use the internet “a couple of times a week” as opposed to “every day” than employed respondents, ceteris paribus. Unemployed and inactive people are more likely not to use the internet at all rather than use it every day, compared with the employed. 
· Older people tend to be more likely not to use the internet at all as opposed to using it every day than younger people, holding other predictors constant. 
· Married and cohabiting people are less likely not to use the internet at all, rather than use it every day, than people not living in a couple, ceteris paribus.

Note that there are fewer significant differences between the first two outcomes, using the internet every day and using it several times a week, than between using it every day and not at all. This is probably because the variable is ordinal (ordered probit or ordered logit may be more intuitive models). 

If we are more interested in the characteristics that make frequent internet use more likely, we can change the base outcome category to “did not use the internet at all” with the option baseoutcome(#):

mlogit internet male i.agegroup i.activity kid edu urban married if country==15 [pw=wgt_target], b(4)

The results can be interpreted using relative risk ratios (the exponents of the coefficients) in a way similar to odds ratios. To report the relative risk ratios, use the rrr option. 

mlogit internet male i.agegroup i.activity kid edu urban married if country==15 [pw=wgt_target], b(4) rrr

Now we can say, for example, that on average: 
· Respondents with secondary education or higher are 6 times more likely to use the internet every day rather than not at all than people with below secondary education, everything else held equal. 
· Unemployed people are only 17% as likely to use the internet every day as employed people, everything else being equal.

ESTIMATING CHANGES IN THE PREDICTED PROBABILITIES
Finally, we can produce the predicted probabilities of each outcome for each person in the dataset using the predict post-estimation command and evaluate the changes in the predicted probabilities resulting from changes in a particular predictor of interest, holding other predictors constant, using margins (Stata 11) or the “clarify” commands (Stata 10 and below).  

To generate the predicted probabilities of each outcome for each person in the estimation sample (if e(sample)) as new variables p1, p2, p3 and p4, use predict after mlogit:

predict p1 p2 p3 p4 if e(sample)
su p1 p2 p3 p4 [aw=wgt_target]

Without the if e(sample) condition, we would generate the predicted probabilities for each person in the entire 31-country dataset, although we’ve estimated our model for Great Britain only. The means of the four new variables sum up to 1 and are exactly the same as the distribution of the internet variable in the estimation sample:

tab internet if e(sample) [aw=wgt_target]

If we are interested in the effect of activity status on the predicted probabilities of using the internet every day, calculate the predicted probabilities for the outcome of interest over four different values of activity, holding all other predictors constant at their zero values for convenience. However, first replace agegroup with continuous age and age squared, then hold age at its mean:

mlogit internet male age c.age#c.age i.activity kid edu urban married if country==15 [pw=wgt_target]
margins activity, atmeans at(male=0 kid=0 edu=0 urban=0 married=0) predict(outcome(1))

In order to clearly see the discrete changes in the probability of the outcome associated with discrete changes in the values of the predictor from its reference category (1), along with their standard errors, using the dydx option:
margins, dydx(activity) atmeans at(male=0 kid=0 edu=0 urban=0 married=0 activity=(1(1)4)) predict(outcome(1))

Thus, among low educated single women without dependent children living in rural areas, those who are unemployed or inactive have significantly lower probabilities of using the internet every day than those who are employed.  Those who are in education or another activity category have significantly higher probabilities of using the internet every day than employed persons. 

Finally, let’s look at the effect of age for the same hypothetical person as above. The only significant effect of age is observed between the two opposite outcomes, using the internet every day (outcome 1) and not using it at all (outcome 4), with younger people being, on average, more likely to use the internet every day rather than not at all (this effect is  non-linear, however). 

margins, at(activity=1 male=0 kid=0 edu=0 urban=0 married=0 age=(18(1)97)) predict(outcome(1))
margins, at(activity=1 male=0 kid=0 edu=0 urban=0 married=0 age=(18(1)97)) predict(outcome(4))

To see the effects of a one unit increase in age on the probabilities of outcomes 1 and 4, holding all predictors at specific values, produce the marginal effects of age with dydx:

margins, dydx(age) at(activity=1 male=0 kid=0 edu=0 urban=0 married=0 age=(18(1)97)) predict(outcome(1)) 
margins, dydx(age) at(activity=1  male=0  kid=0 edu=0 urban=0 married=0 age=(18(1)97)) predict(outcome(4))

EXERCISE 1
Estimate a simple multinomial logit model of current activity status as a function of gender, marital status, presence of children, age and age squared, education dummy, and health status in a country of your choice. Then include a 3-way interaction between gender, marital status and having dependent children. Which characteristics are associated with an increased likelihood of being employed? Interpret the interaction effect between gender, marital status and having dependent children, if any. For a hypothetical person of your choice, produce the differences in the predicted probabilities of each outcome for persons with and without secondary education. For which outcomes are these differences statistically significant?   

SURVIVAL ANALYSIS: GETTING DATA INTO THE DESIRED FORMAT
To model the duration of an episode, of the length of time that units spend in some state (e.g. unemployment) before experiencing an event of interest (e.g. transition to employment) it is often necessary to transform the data into an appropriate format. Open “BHPS unemployment spells.dta” containing a person-period activity history file and produce a list of the first 20 or so observations for the variables episno (episode number), empstat (activity status), epdate (start date of the episode) and epend (end date of the episode), separating the output by personal identifiers (pid):

list episno empstat  epdate epend in 1/20, sepby(pid)

An activity episode may contain several spells, so we need to aggregate the data by person-episodes, so that each record contains a separate person-episode. First drop any observations with missing activity status and recode the activity status into a dummy (employed/non-employed):

drop if empstat<0
recode empstat (1=1 employed) (2/10=0 "unempl/inactive") (else=.), into(employed)
collapse (first) empstat  epdate epend epcens age  male edu mstat country employed , by(pid episno)
list episno empstat  epdate epend in 1/20, sepby(pid)

Since value labels have been lost during data aggregation, re-label the variables of interest:

lab val empstat empstat
lab val mstat mstat
lab val edu edu
lab val country country 

Drop any variables with missing values on either the start or end date of the episode, as well as any observations where the end date precedes the start date.

drop if epdate==-9 | epend==-9
drop if epend<epdate

Now we can create the dependent variable: episode duration or the length of time that a person spends at risk of an event between the onset of risk (at time zero) and the time of failure.  

gen epdur=(epend-epdate)+1

The next most important variable is the identifier of whether the event happened at the end of the episode or not. Create a dummy variable failure that takes on the value of 1 if a person who is not employed at time t is employed at time t+1:

gen failure=0
bysort pid: replace failure=1 if employed[_n]==0 & employed[_n+1]==1

Now we are ready to tell Stata about the structure of our survival data using the command stset. We need to “stset” the data only once before estimating any number of duration models. We need to tell Stata which variables denote the time of failure (epdur) and the failure indicator itself (failure). Moreover, we need to tell Stata that there may be several records for one individual, using the option id; that a person who experienced the event of interest may re-enter the risk set again if, for example, a non-employed person becomes employed and then unemployed again, using the option exit(time .);  and that individuals enter the risk set when they are not employed, using the option enter:

stset epdur, failure(failure)  enter(employed==0) exit(time .) id(pid)

To get some basic descriptive survival statistics, use stsum and stdes.  
stsum
stdes
The incidence rate produced by stsum is the number of ‘failures’ over the total time at risk.
Four new variables have been produced by stset: 
_t0: start of the episode
_t: end of the episode
_d: outcome at the end of the episode 
_st: indicator of whether the observation is relevant in the current analysis

SURVIVAL ANALYSIS: ESTIMATION
To get an idea of what the survival function for the data defined with stset looks like:

sts graph
sts graph if _t<200
sts graph if _t<200, by(male)

Lets estimate a log relative hazard model with a few simple demographic predictors. At first, include only the male dummy, age and age squared; then add education, marital status and region. 

gen agesq=age^2
streg male age agesq, distribution(exponential) nohr 

The constant is the estimate of the baseline hazard which is assumed to be constant.  There significant differences in the hazard rate between men and women, as men are less likely to make the transition to employment, controlling for age. Older people are more likely to make the transition, although the positive effect of age eventually wears off.

streg male age agesq i.edu i.mstat i.country , distribution(exponential) nohr 

The effect of education, as would be expected, is to increase the hazard of entering employment, although there are no significant differences between people with university education and those with A levels only, everything else being equal. 

Estimate an accelerated time failure model with the same predictors. It is a mirror image of the log relative hazard model.

streg male age agesq, distribution(exponential) nohr time
streg male age agesq i.edu i.mstat i.country , distribution(exponential) nohr time

The effect of age is to accelerate failure (the transition to employment) initially and then, eventually, to delay failure (as the significantly negative quadratic term indicates). The effect of being male is to delay failure. That is, the model suggests that men spend longer in non-employment and women make the transition to employment faster.  

One of the biggest limitations of this analysis is that long-term inactive people have been grouped together with the short-term unemployed. 

